Store the original Smith and Wagner estimators and restate them in terms of estimated probability (ep) instead of n; Specify the gain estimator

In[21]:=
```
mu = Simplify[enl + erl + (-1 + enl + erl)/(-1 + n) /. n → 1 / ep]

gamma = Simplify[n (-1 + enl + erl + epl n)/(-1 + n)^2 /. n → 1 / ep]

alpha = Simplify[n (-1 + epl + erl + enl n)/(-1 + n)^2 /. n → 1 / ep]

gain = Simplify[gamma/(1 - mu)]
```

Out[21]=
$$\frac{enl - ep + erl}{1 - ep}$$

Out[22]=
$$\frac{epl + ep\,(-1 + enl + erl)}{(-1 + ep)^2}$$

Out[23]=
$$\frac{enl + ep\,(-1 + epl + erl)}{(-1 + ep)^2}$$

Out[24]=
$$\frac{epl + ep\,(-1 + enl + erl)}{(-1 + ep)\,(-1 + enl + erl)}$$

Solve for implied probability when alpha =0; Substitute the value into the gamma and gain estimator

In[25]:=
```
nls = Flatten[Solve[alpha == 0, ep]]
```

Out[25]=
$$\left\{ep \to -\frac{enl}{-1 + epl + erl}\right\}$$

In[26]:=
```
gammaZ = Simplify[gamma /. nls]
gainZ = Simplify[gain /. nls]
```

Out[26]=
$$-\frac{(enl - epl)\,(-1 + epl + erl)}{-1 + enl + epl + erl}$$

Out[27]=
$$\frac{enl - epl}{-1 + enl + erl}$$

Restate estimators in terms of p and Δ

In[28]:=
```
gammaE = gamma /. ep → p + Δ
gainE = gain /. ep → p + Δ
```

Out[28]=
$$\frac{epl + (-1 + enl + erl)\ (p + \Delta)}{(-1 + p + \Delta)^2}$$

Out[29]=
$$\frac{epl + (-1 + enl + erl)\ (p + \Delta)}{(-1 + enl + erl)\ (-1 + p + \Delta)}$$

Calculate the elasticity of the two estimators

In[30]:=
```
ElasGain = Simplify[D[gainE, Δ] Δ/gainE]
```

Out[30]=
$$-\frac{(-1 + enl + epl + erl)\ \Delta}{(-1 + p + \Delta)\ (epl + (-1 + enl + erl)\ (p + \Delta))}$$

In[31]:=
```
ElasGamma = Simplify[D[gammaE, Δ] Δ/gammaE]
```

Out[31]=
$$-\frac{\Delta\ (2\ epl + enl\ (1 + p + \Delta) + (-1 + erl)\ (1 + p + \Delta))}{(-1 + p + \Delta)\ (epl + (-1 + enl + erl)\ (p + \Delta))}$$

Calculate the ratio of the elasticities

In[32]:=
```
R = FullSimplify[ElasGain/ElasGamma]
```

Out[32]=
$$\frac{-1 + enl + epl + erl}{2\ epl + enl\ (1 + p + \Delta) + (-1 + erl)\ (1 + p + \Delta)}$$

Calculate the ratio of the elasticities given alpha = 0

In[34]:=
```
RA = Simplify[R /. p + Δ → ep /. nls]
```

Out[34]=
$$-\frac{-1 + epl + erl}{1 + enl - 2\ epl - erl}$$